// File: contracts/LilVillains.sol



// SPDX-License-Identifier: CC-BY-NC-ND-4.0
// ”I have read and accept the Terms of service (link: https://lilheroes.io/tos) and privacy policy (https://lilheroes.io/privacy)”

pragma solidity ^0.8.9;

import '@openzeppelin/contracts/access/AccessControl.sol';
import '@openzeppelin/contracts/access/Ownable.sol';
import '@gm2/blockchain/src/contracts/GM721.sol';
import '@gm2/blockchain/src/contracts/GM2981.sol';
import { TransferIsNotSupported } from '@gm2/blockchain/src/errors/GM721Errors.sol';
import { Royalty } from '@gm2/blockchain/src/structs/DynamicMetadataStructs.sol';
import './common/contracts/GMDynamicSC.sol';
import './interfaces/ILilCollection.sol';
import './errors/LilVillainsErrors.sol';

contract LilVillains is GM2981, AccessControl, GM721, GMDynamicSC, Ownable, ILilCollection {
  bytes32 public constant MINTER_ROLE = keccak256('MINTER_ROLE');
  bytes32 public constant ALTER_ATTR_ROLE = keccak256('ALTER_ATTR_ROLE');
  bytes32 public constant ALTER_IMAGE_ROLE = keccak256('ALTER_IMAGE_ROLE');
  uint32 private constant MAX_SUPPLY = 7777;
  bool revealed = false;
  string public contractURI;

  string[10] traitTypes = ['Back', 'Mouth', 'Clothes', 'Ears', 'Helmet', 'Nose', 'Eyes', 'Accessory', 'Hat', 'Skin'];

  constructor(
    address villainMinter,
    address royaltyAddressToSet,
    string memory baseURI,
    string memory collectionDescription,
    string memory contractURI_
  ) GM721('LilVillains', 'LILV') GM2981(royaltyAddressToSet, 800) GMDynamicSC(baseURI, collectionDescription) {
    //8% of royalty
    contractURI = contractURI_;
    _grantRole(DEFAULT_ADMIN_ROLE, msg.sender);
    _grantRole(ALTER_ATTR_ROLE, villainMinter);
    _grantRole(ALTER_IMAGE_ROLE, msg.sender);
    _grantRole(MINTER_ROLE, villainMinter);
  }

  modifier onlyTokenExists(uint256 tokenId) {
    if (!_exists(tokenId)) {
      revert TokenDoesNotExists(tokenId);
    }
    _;
  }

  function setContractURI(string calldata newContractURI) external onlyRole(DEFAULT_ADMIN_ROLE) {
    contractURI = newContractURI;
  }

  function setBaseURI(string calldata newBaseUri) external onlyRole(DEFAULT_ADMIN_ROLE) {
    if (revealed) {
      revert CollectionAlreadyRevealed();
    }
    _defaultBaseURI = newBaseUri;
    revealed = true;
  }

  function batchMint(address to, uint256[] calldata tokenIds) external onlyRole(MINTER_ROLE) {
    super._batchMint(to, tokenIds);
  }

  function setBaseAttributes(NFTBaseAttributes[] calldata nFTsBaseAttributes) external onlyRole(ALTER_ATTR_ROLE) {
    for (uint256 i = 0; i < nFTsBaseAttributes.length; i++) {
      NFTBaseAttributes memory nFTBaseAttributes = nFTsBaseAttributes[i];
      Attribute[] memory baseAttributes = new Attribute[](nFTBaseAttributes.values.length);
      for (uint256 j = 0; j < baseAttributes.length; j++) {
        baseAttributes[j] = Attribute('', traitTypes[j], nFTBaseAttributes.values[j]);
      }
      _setBaseAttributes(nFTBaseAttributes.id, baseAttributes);
    }
  }

  function setRoyaltyAddress(address newRoyaltyAddress) external onlyRole(DEFAULT_ADMIN_ROLE) {
    _royaltyAddress = newRoyaltyAddress;
  }

  function tokenURI(uint256 tokenId) public view virtual override onlyTokenExists(tokenId) returns (string memory) {
    return sCBehavior.getTokenURI(tokenId);
  }

  function getDynamicSCAddresses() public view override onlyRole(DEFAULT_ADMIN_ROLE) returns (address[2] memory) {
    return super.getDynamicSCAddresses();
  }

  function setDefaultImagePath(string calldata newDefaultImagePath) public onlyRole(DEFAULT_ADMIN_ROLE) {
    _defaultImagePath = newDefaultImagePath;
  }

  function setAttributesControllerSC(address newAttributesControllerSC) public onlyRole(DEFAULT_ADMIN_ROLE) {
    _setAttributesControllerSC(newAttributesControllerSC);
  }

  function setTransferControllerSC(address newTransferControllerSC) public onlyRole(DEFAULT_ADMIN_ROLE) {
    _setTransferControllerSC(newTransferControllerSC);
  }

  function setImagePath(uint256 tokenId, string memory newImageURL) public onlyRole(ALTER_IMAGE_ROLE) {
    _setImagePath(tokenId, newImageURL);
  }

  function supportsInterface(bytes4 interfaceId) public view override(GM2981, AccessControl, ERC721) returns (bool) {
    return interfaceId == type(ILilCollection).interfaceId || super.supportsInterface(interfaceId);
  }

  function transferOwnership(address newOwner)
    public
    override
    onlyRole(DEFAULT_ADMIN_ROLE)
    onlyNotZeroAddress(newOwner)
  {
    super.transferOwnership(newOwner);
    grantRole(DEFAULT_ADMIN_ROLE, newOwner);
    revokeRole(DEFAULT_ADMIN_ROLE, msg.sender);
  }

  function _beforeBatchMint(address, uint256[] calldata tokenIds) internal view override(GM721) {
    uint256 currentSupply = totalSupply();
    uint256 supplyRequired = currentSupply + tokenIds.length;
    if (supplyRequired > MAX_SUPPLY) {
      revert SupplyIsNotEnought(tokenIds.length, currentSupply, MAX_SUPPLY);
    }
  }

  function _beforeRoyaltyInfo(uint256 tokenId) internal view override(GM2981) onlyTokenExists(tokenId) {}

  function _beforeTokenTransfer(
    address from,
    address to,
    uint256 tokenId
  ) internal view override(ERC721) {
    if (!sCBehavior.canTokenBeTransferred(from, to, tokenId)) {
      revert TokenIsBlockedToTransfer(tokenId);
    }
  }

  function _getRoyaltiesData() internal view override returns (Royalty memory) {
    return Royalty(_royaltyAddress, _royaltyPercentage);
  }

  function _beforeTransferBlockedToken(address to, uint256 tokenId) internal override {
    if (!_checkERC721Compatibility(msg.sender, to, tokenId, '')) {
      revert TransferIsNotSupported(to, tokenId);
    }
  }

  function _transferBlockedToken(address to, uint256 tokenId) internal override {
    _transfer(msg.sender, to, tokenId);
  }
}

// File: @openzeppelin/contracts/access/AccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (access/AccessControl.sol)

pragma solidity ^0.8.0;

import "./IAccessControl.sol";
import "../utils/Context.sol";
import "../utils/Strings.sol";
import "../utils/introspection/ERC165.sol";

/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms. This is a lightweight version that doesn't allow enumerating role
 * members except through off-chain means by accessing the contract event logs. Some
 * applications may benefit from on-chain enumerability, for those cases see
 * {AccessControlEnumerable}.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it.
 */
abstract contract AccessControl is Context, IAccessControl, ERC165 {
    struct RoleData {
        mapping(address => bool) members;
        bytes32 adminRole;
    }

    mapping(bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Modifier that checks that an account has a specific role. Reverts
     * with a standardized message including the required role.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     *
     * _Available since v4.1._
     */
    modifier onlyRole(bytes32 role) {
        _checkRole(role, _msgSender());
        _;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControl).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view virtual override returns (bool) {
        return _roles[role].members[account];
    }

    /**
     * @dev Revert with a standard message if `account` is missing `role`.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     */
    function _checkRole(bytes32 role, address account) internal view virtual {
        if (!hasRole(role, account)) {
            revert(
                string(
                    abi.encodePacked(
                        "AccessControl: account ",
                        Strings.toHexString(uint160(account), 20),
                        " is missing role ",
                        Strings.toHexString(uint256(role), 32)
                    )
                )
            );
        }
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view virtual override returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been revoked `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) public virtual override {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     *
     * NOTE: This function is deprecated in favor of {_grantRole}.
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        bytes32 previousAdminRole = getRoleAdmin(role);
        _roles[role].adminRole = adminRole;
        emit RoleAdminChanged(role, previousAdminRole, adminRole);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * Internal function without access restriction.
     */
    function _grantRole(bytes32 role, address account) internal virtual {
        if (!hasRole(role, account)) {
            _roles[role].members[account] = true;
            emit RoleGranted(role, account, _msgSender());
        }
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * Internal function without access restriction.
     */
    function _revokeRole(bytes32 role, address account) internal virtual {
        if (hasRole(role, account)) {
            _roles[role].members[account] = false;
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @gm2/blockchain/src/contracts/GM721.sol



// SPDX-License-Identifier: MIT
// GM2 Contracts (last updated v0.0.1)

pragma solidity ^0.8.9;

import '@openzeppelin/contracts/token/ERC721/ERC721.sol';
import '@openzeppelin/contracts/token/ERC721/extensions/ERC721Burnable.sol';
import '@openzeppelin/contracts/utils/Address.sol';
import '../errors/GM721Errors.sol';

contract GM721 is ERC721, ERC721Burnable {
  using Address for address;
  uint256 internal _supply = 0;

  constructor(string memory name_, string memory symbol_) ERC721(name_, symbol_) {}

  function totalSupply() public view returns (uint256) {
    return _supply;
  }

  function _batchMint(address to, uint256[] calldata tokenIds) internal {
    _beforeBatchMint(to, tokenIds);

    uint256 tokensToMint = tokenIds.length;

    _supply += tokensToMint;
    for (uint256 index = 0; index < tokensToMint; index = _increment(index)) {
      _safeMint(to, tokenIds[index]);
    }
  }

  function burn(uint256 tokenId) public override {
    super.burn(tokenId);
    _supply--;
  }

  function _beforeBatchMint(address, uint256[] calldata) internal virtual {}

  function _increment(uint256 i) internal pure returns (uint256) {
    return i = i + 1;
  }

  function _checkERC721Compatibility(
    address from,
    address to,
    uint256 tokenId,
    bytes memory _data
  ) internal returns (bool) {
    if (to.isContract()) {
      try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
        return retval == IERC721Receiver.onERC721Received.selector;
      } catch (bytes memory reason) {
        if (reason.length == 0) {
          revert TransferIsNotSupported(to, tokenId);
        } else {
          assembly {
            revert(add(32, reason), mload(reason))
          }
        }
      }
    } else {
      return true;
    }
  }
}

// File: @gm2/blockchain/src/contracts/GM2981.sol



// SPDX-License-Identifier: MIT
// GM2 Contracts (last updated v0.0.1)

pragma solidity ^0.8.9;

import '@openzeppelin/contracts/interfaces/IERC2981.sol';
import '@openzeppelin/contracts/utils/introspection/ERC165.sol';

contract GM2981 is IERC2981, ERC165 {
  address internal _royaltyAddress;
  uint16 internal _royaltyPercentage;

  constructor(address royaltyAddress_, uint16 royaltyPercentage_) {
    require(royaltyAddress_ != address(0), 'Invalid royalty address');
    require(royaltyPercentage_ <= 2000, 'Invalid royalty percentage(MAX:2000)');
    _royaltyAddress = royaltyAddress_;
    _royaltyPercentage = royaltyPercentage_;
  }

  function royaltyInfo(uint256 tokenId, uint256 salePrice)
    external
    view
    override
    returns (address receiver, uint256 royaltyAmount)
  {
    _beforeRoyaltyInfo(tokenId);
    receiver = _royaltyAddress;
    royaltyAmount = _calculatePercentage(salePrice, _royaltyPercentage);
  }

  function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
    return interfaceId == type(IERC2981).interfaceId || super.supportsInterface(interfaceId);
  }

  function _beforeRoyaltyInfo(uint256) internal view virtual {}

  function _calculatePercentage(uint256 value, uint16 percent) internal pure returns (uint256) {
    return (value * percent) / 10000; //HUNDRED PERCENT is 10000
  }
}

// File: @gm2/blockchain/src/errors/GM721Errors.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;
// Invalid operation. To doesn't implements IERC721Receiver(to).onERC721Received
// @param to to address.
// @param tokenId sent tokenId.
error TransferIsNotSupported(address to, uint256 tokenId);

// File: @gm2/blockchain/src/structs/DynamicMetadataStructs.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;
struct Attribute {
  string displayType;
  string traitType;
  string value;
}

struct Royalty {
  address recipientAddress;
  uint16 feePercentage; // INFO: Use two decimal => 100 = 1%
}

struct Metadata {
  string description;
  string name;
  Tuple[] additionalProperties;
  Attribute[] attributes;
}

struct Tuple {
  string key;
  string value;
}

struct SCBehavior {
  function(uint256) internal view returns (string memory) getTokenURI;
  function(address, address, uint256) internal view returns (bool) canTokenBeTransferred;
  function(address, uint256) internal transferBlockedToken;
}

// File: contracts/common/contracts/GMDynamicSC.sol



// SPDX-License-Identifier: MIT
// GM2 Contracts (last updated v0.0.1)
pragma solidity ^0.8.9;

import '@openzeppelin/contracts/utils/introspection/IERC165.sol';
import '@openzeppelin/contracts/utils/Strings.sol';
import '@gm2/blockchain/src/interfaces/IGMTransferController.sol';
import '@gm2/blockchain/src/interfaces/IGMAttributesController.sol';
import '@gm2/blockchain/src/errors/AddressValidatorErrors.sol';
import { Attribute, Royalty, SCBehavior } from '@gm2/blockchain/src/structs/DynamicMetadataStructs.sol';
import { Metadata as MetadataV1 } from '../structs/DynamicMetadataStructs.sol';

import '../libraries/DynamicMetadata.sol';
import '../errors/DynamicMetadataErrors.sol';

abstract contract GMDynamicSC {
  using Strings for uint256;
  using DynamicMetadata for Attribute[];
  using DynamicMetadata for MetadataV1;

  mapping(uint256 => Attribute[]) _baseAttributesDictionary;
  mapping(uint256 => string) _imageURLDictionary;
  address _attributesControllerSC;
  address _transferControllerSC;
  string _defaultBaseURI;
  string _defaultImagePath;
  string _metadataDescription;

  SCBehavior sCBehavior = SCBehavior(_defaultTokenURI, _defaultCanTokenBeTransferred, _defaultTransferBlockedToken);

  constructor(string memory baseURI, string memory description) {
    _defaultBaseURI = baseURI;
    _metadataDescription = description;
  }

  modifier onlyNotZeroAddress(address addr) {
    if (addr == address(0)) {
      revert ZeroAddressNotSupported();
    }
    _;
  }

  modifier onlyIfSupports(address addr, bytes4 interfaceId) {
    if (IERC165(addr).supportsInterface(interfaceId) == false) {
      revert InterfaceIsNotImplemented(addr, interfaceId);
    }
    _;
  }

  function transferBlockedToken(address to, uint256 tokenId) public onlyNotZeroAddress(to) {
    sCBehavior.transferBlockedToken(to, tokenId);
  }

  function getDynamicSCAddresses() public view virtual returns (address[2] memory) {
    return [_attributesControllerSC, _transferControllerSC];
  }

  function _setAttributesControllerSC(address newAttributesControllerSC)
    internal
    virtual
    onlyNotZeroAddress(newAttributesControllerSC)
    onlyIfSupports(newAttributesControllerSC, type(IGMAttributesController).interfaceId)
  {
    _beforeSetAttributesControllerSC(newAttributesControllerSC);
    _attributesControllerSC = newAttributesControllerSC;
    sCBehavior.getTokenURI = _getDynamicTokenURI;
  }

  function _beforeSetAttributesControllerSC(address) internal virtual {
    if (bytes(_defaultImagePath).length == 0) {
      revert DefaultImagePathRequired();
    }
  }

  function _setTransferControllerSC(address newTransferControllerSC)
    internal
    virtual
    onlyNotZeroAddress(newTransferControllerSC)
    onlyIfSupports(newTransferControllerSC, type(IGMTransferController).interfaceId)
  {
    _beforeSetTransferControllerSC(newTransferControllerSC);
    _transferControllerSC = newTransferControllerSC;
    sCBehavior.canTokenBeTransferred = _dynamicCanTokenBeTransferred;
    sCBehavior.transferBlockedToken = _dynamicTransferBlockedToken;
  }

  function _beforeSetTransferControllerSC(address) internal virtual {}

  function _setImagePath(uint256 tokenId, string memory newImageURL) internal virtual {
    _beforeSetImagePath(tokenId, newImageURL);
    _imageURLDictionary[tokenId] = newImageURL;
  }

  function _beforeSetImagePath(uint256 tokenId, string memory newImageURL) internal virtual {
    if (_baseAttributesDictionary[tokenId].length == 0) {
      revert CannotSetImageWithoutBaseAttributes(tokenId, newImageURL);
    }
  }

  function _setBaseAttributes(uint256 tokenId, Attribute[] memory newBaseAttributes) internal virtual {
    _beforeSetBaseAttributes(tokenId, newBaseAttributes);
    _baseAttributesDictionary[tokenId].appendBaseAttributes(newBaseAttributes);
  }

  function _beforeSetBaseAttributes(uint256 tokenId, Attribute[] memory) internal virtual {
    if (_baseAttributesDictionary[tokenId].length != 0) {
      revert AlreadyHaveBaseAttributes(tokenId);
    }
  }

  function _dynamicTransferBlockedToken(address to, uint256 tokenId) internal {
    _beforeTransferBlockedToken(to, tokenId);
    IGMTransferController(_transferControllerSC).bypassTokenId(address(this), tokenId);
    _transferBlockedToken(to, tokenId);
    IGMTransferController(_transferControllerSC).removeBypassTokenId(address(this), tokenId);
  }

  function _beforeTransferBlockedToken(address to, uint256 tokenId) internal virtual {}

  function _transferBlockedToken(address to, uint256 tokenId) internal virtual {}

  function _getRoyaltiesData() internal view virtual returns (Royalty memory) {}

  function _getDynamicTokenURI(uint256 tokenId) internal view returns (string memory) {
    string memory tokenURI = _defaultTokenURI(tokenId);
    Attribute[] memory attributes = _baseAttributesDictionary[tokenId];

    // INFO: Has Base attributes
    if (attributes.length != 0) {
      Attribute[] memory dynamicAttributes = IGMAttributesController(_attributesControllerSC).getDynamicAttributes(
        address(this),
        tokenId
      );

      // INFO: Has dynamic attributes
      if (dynamicAttributes.length > 0) {
        attributes = _baseAttributesDictionary[tokenId].concatDynamicAttributes(dynamicAttributes);
      }

      string memory metadataImageURL = _getImageURL(tokenId);
      string memory metadataName = string(abi.encodePacked('#', tokenId.toString()));
      Royalty memory royalty = _getRoyaltiesData();

      MetadataV1 memory metadata = MetadataV1(
        _metadataDescription,
        metadataImageURL,
        metadataName,
        attributes,
        royalty
      );

      tokenURI = metadata.toBase64URI();
    }

    return tokenURI;
  }

  // INFO: Behavior dynamic methods
  function _dynamicCanTokenBeTransferred(
    address from,
    address to,
    uint256 tokenId
  ) private view returns (bool) {
    return IGMTransferController(_transferControllerSC).canTokenBeTransferred(address(this), from, to, tokenId);
  }

  function _getImageURL(uint256 tokenId) private view returns (string memory) {
    string memory imagePath = string(abi.encodePacked(_defaultImagePath, tokenId.toString()));
    if (bytes(_imageURLDictionary[tokenId]).length > 0) {
      // INFO: TokenID should be in the custom image path
      imagePath = _imageURLDictionary[tokenId];
    }
    return imagePath;
  }

  // INFO: Default behavior methods
  function _defaultTokenURI(uint256 tokenId) private view returns (string memory) {
    return string(abi.encodePacked(_defaultBaseURI, tokenId.toString()));
  }

  function _defaultCanTokenBeTransferred(
    address,
    address,
    uint256
  ) private pure returns (bool) {
    return true;
  }

  function _defaultTransferBlockedToken(address, uint256) private pure {
    revert MethodNotSupported('TransferBlockedToken');
  }
}

// File: contracts/interfaces/ILilCollection.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;

import { NFTBaseAttributes } from '../structs/LilVillainsStructs.sol';

interface ILilCollection {
  function batchMint(address to, uint256[] calldata tokenIds) external;

  function setBaseAttributes(NFTBaseAttributes[] memory nFTsBaseAttributes) external;
}

// File: contracts/errors/LilVillainsErrors.sol



// SPDX-License-Identifier: CC-BY-NC-ND-4.0
pragma solidity ^0.8.9;

// Invalid operation. Collection was already revealed
error CollectionAlreadyRevealed();
// Invalid operation. Token does not exists
// @param tokenId sent tokenId.
error TokenDoesNotExists(uint256 tokenId);
// Invalid operation. Not enought supply
// @param amountOfTokensToMint sent amount of tokensToMint.
// @param currentSupply currentSupply.
// @param maxSupportedSupply maxSupportedSupply.
error SupplyIsNotEnought(uint256 amountOfTokensToMint, uint256 currentSupply, uint256 maxSupportedSupply);
// Invalid operation. Token can't be transfered because is blocked
// @param tokenId sent tokenId.
error TokenIsBlockedToTransfer(uint256 tokenId);

// File: @openzeppelin/contracts/access/IAccessControl.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

pragma solidity ^0.8.0;

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControl {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)

pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: @openzeppelin/contracts/token/ERC721/ERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC721/ERC721.sol)

pragma solidity ^0.8.0;

import "./IERC721.sol";
import "./IERC721Receiver.sol";
import "./extensions/IERC721Metadata.sol";
import "../../utils/Address.sol";
import "../../utils/Context.sol";
import "../../utils/Strings.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping(uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping(address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping(uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC721).interfaceId ||
            interfaceId == type(IERC721Metadata).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0 ? string(abi.encodePacked(baseURI, tokenId.toString())) : "";
    }

    /**
     * @dev Base URI for computing {tokenURI}. If set, the resulting URI for each
     * token will be the concatenation of the `baseURI` and the `tokenId`. Empty
     * by default, can be overriden in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(
            _msgSender() == owner || isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(
        address to,
        uint256 tokenId,
        bytes memory _data
    ) internal virtual {
        _mint(to, tokenId);
        require(
            _checkOnERC721Received(address(0), to, tokenId, _data),
            "ERC721: transfer to non ERC721Receiver implementer"
        );
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);

        _afterTokenTransfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);

        _afterTokenTransfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer from incorrect owner");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);

        _afterTokenTransfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits a {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC721: approve to caller");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(
        address from,
        address to,
        uint256 tokenId,
        bytes memory _data
    ) private returns (bool) {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver.onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 tokenId
    ) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC721/extensions/ERC721Burnable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/ERC721Burnable.sol)

pragma solidity ^0.8.0;

import "../ERC721.sol";
import "../../../utils/Context.sol";

/**
 * @title ERC721 Burnable Token
 * @dev ERC721 Token that can be irreversibly burned (destroyed).
 */
abstract contract ERC721Burnable is Context, ERC721 {
    /**
     * @dev Burns `tokenId`. See {ERC721-_burn}.
     *
     * Requirements:
     *
     * - The caller must own `tokenId` or be an approved operator.
     */
    function burn(uint256 tokenId) public virtual {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721Burnable: caller is not owner nor approved");
        _burn(tokenId);
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/token/ERC721/IERC721.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external;
}

// File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/IERC721Receiver.sol)

pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC721/extensions/IERC721Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC721.sol";

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {
    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}

// File: @openzeppelin/contracts/interfaces/IERC2981.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (interfaces/IERC2981.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Interface for the NFT Royalty Standard.
 *
 * A standardized way to retrieve royalty payment information for non-fungible tokens (NFTs) to enable universal
 * support for royalty payments across all NFT marketplaces and ecosystem participants.
 *
 * _Available since v4.5._
 */
interface IERC2981 is IERC165 {
    /**
     * @dev Returns how much royalty is owed and to whom, based on a sale price that may be denominated in any unit of
     * exchange. The royalty amount is denominated and should be payed in that same unit of exchange.
     */
    function royaltyInfo(uint256 tokenId, uint256 salePrice)
        external
        view
        returns (address receiver, uint256 royaltyAmount);
}

// File: @openzeppelin/contracts/interfaces/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (interfaces/IERC165.sol)

pragma solidity ^0.8.0;

import "../utils/introspection/IERC165.sol";

// File: @gm2/blockchain/src/interfaces/IGMTransferController.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;

import '@openzeppelin/contracts/utils/introspection/IERC165.sol';
import '@openzeppelin/contracts-upgradeable/access/IAccessControlUpgradeable.sol';

interface IGMTransferController is IERC165, IAccessControlUpgradeable {
  function canTokenBeTransferred(
    address collectionAddress,
    address from,
    address to,
    uint256 tokenId
  ) external view returns (bool);

  function bypassTokenId(address collectionAddress, uint256 tokenId) external;

  function removeBypassTokenId(address collectionAddress, uint256 tokenId) external;
}

// File: @gm2/blockchain/src/interfaces/IGMAttributesController.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;

import '@openzeppelin/contracts/utils/introspection/IERC165.sol';
import { Attribute } from '../structs/DynamicMetadataStructs.sol';

interface IGMAttributesController is IERC165 {
  function getDynamicAttributes(address from, uint256 tokenId) external view returns (Attribute[] memory attributes);
}

// File: @gm2/blockchain/src/errors/AddressValidatorErrors.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;

// Invalid address sent. It doesn't implements the require interface
// @param addressSent sent address.
// @param interfaceRequired required interface.
error InterfaceIsNotImplemented(address addressSent, bytes4 interfaceRequired);
// Invalid address sent. Needed a not zero address
error ZeroAddressNotSupported();

// File: contracts/common/structs/DynamicMetadataStructs.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;

import { Attribute, Royalty, SCBehavior } from '@gm2/blockchain/src/structs/DynamicMetadataStructs.sol';

struct Metadata {
  string description;
  string image;
  string name;
  Attribute[] attributes;
  Royalty royalty;
}

// File: contracts/common/libraries/DynamicMetadata.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;

import '@openzeppelin/contracts/utils/Base64.sol';
import '@openzeppelin/contracts/utils/Strings.sol';
import { Attribute, Royalty } from '@gm2/blockchain/src/structs/DynamicMetadataStructs.sol';
import { Metadata as MetadataV1 } from '../structs/DynamicMetadataStructs.sol';

library DynamicMetadata {
  using Strings for uint16;

  // Returns a copy of the concated attributes
  function concatDynamicAttributes(Attribute[] calldata baseAttributes, Attribute[] calldata dynamicAttributes)
    public
    pure
    returns (Attribute[] memory)
  {
    uint256 countOfAttributes = baseAttributes.length + dynamicAttributes.length;
    Attribute[] memory allAttributes = new Attribute[](countOfAttributes);
    for (uint256 i = 0; i < baseAttributes.length; i++) {
      allAttributes[i] = baseAttributes[i];
    }

    for (uint256 i = 0; i < dynamicAttributes.length; i++) {
      allAttributes[baseAttributes.length + i] = dynamicAttributes[i];
    }

    return allAttributes;
  }

  //
  function appendBaseAttributes(Attribute[] storage baseAttributes, Attribute[] calldata newBaseAttributes) public {
    for (uint256 i = 0; i < newBaseAttributes.length; i++) {
      baseAttributes.push(newBaseAttributes[i]);
    }
  }

  function toBytes(Attribute memory attribute) public pure returns (bytes memory) {
    bytes memory attributesInByte = '{';
    if (bytes(attribute.displayType).length > 0) {
      attributesInByte = bytes.concat(
        attributesInByte,
        abi.encodePacked('"display_type":"', attribute.displayType, '",')
      );
    }

    attributesInByte = bytes.concat(
      attributesInByte,
      //TODO: ver como hacemos con el manejo de las comillas para los numeros
      abi.encodePacked('"trait_type":"', attribute.traitType, '",', '"value":"', attribute.value, '"')
    );

    return bytes.concat(attributesInByte, '}');
  }

  //INFO: This method assumes that attributesToMap has a length greather than 0
  function mapToBytes(Attribute[] calldata attributesToMap) public pure returns (bytes memory) {
    bytes memory attributesInBytes = '"attributes":[';

    for (uint32 i = 0; i < attributesToMap.length - 1; i++) {
      attributesInBytes = bytes.concat(attributesInBytes, toBytes(attributesToMap[i]), ',');
    }
    attributesInBytes = bytes.concat(attributesInBytes, toBytes(attributesToMap[attributesToMap.length - 1]), ']');

    return attributesInBytes;
  }

  //INFO: This method assumes that metadata.attributes has a length greather than 0
  function toBase64URI(MetadataV1 calldata metadata) public pure returns (string memory) {
    bytes memory descriptionInBytes = keyValueToJsonInBytes('description', metadata.description);
    bytes memory imageInBytes = keyValueToJsonInBytes('image', metadata.image);
    bytes memory nameInBytes = keyValueToJsonInBytes('name', metadata.name);
    bytes memory attributesInBytes = mapToBytes(metadata.attributes);
    bytes memory jsonInBytes = bytes.concat(
      '{',
      descriptionInBytes,
      ',',
      imageInBytes,
      ',',
      nameInBytes,
      ',',
      attributesInBytes
    );
    // INFO: If has royalties set
    if (metadata.royalty.feePercentage != 0) {
      bytes memory royaltyFeeInBytes = keyValueToJsonInBytes(
        'seller_fee_basis_points', // INFO: Open see key for royalties
        metadata.royalty.feePercentage.toString()
      );
      bytes memory royaltyRecipientInBytes = keyValueToJsonInBytes(
        'fee_recipient', // INFO: Open see key for royalties
        Strings.toHexString(uint256(uint160(metadata.royalty.recipientAddress)), 20)
      );
      jsonInBytes = bytes.concat(jsonInBytes, ',', royaltyFeeInBytes, ',', royaltyRecipientInBytes);
    }

    jsonInBytes = bytes.concat(jsonInBytes, '}');

    return string(abi.encodePacked('data:application/json;base64,', Base64.encode(jsonInBytes)));
  }

  function keyValueToJsonInBytes(string memory key, string memory value) public pure returns (bytes memory) {
    return bytes.concat('"', bytes(key), '":"', bytes(value), '"');
  }
}

// File: contracts/common/errors/DynamicMetadataErrors.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;
// Invalid operation. Needed _defaultImagePath to be set
error DefaultImagePathRequired();
// Invalid operation. Can't set imagePath to token without baseAttributes'
// @param tokenId sent tokenId.
// @param imagePath sent imagePath.
error CannotSetImageWithoutBaseAttributes(uint256 tokenId, string imagePath);
// Invalid operation. Token already has base attributes set.
// @param tokenId sent tokenId.
error AlreadyHaveBaseAttributes(uint256 tokenId);
// Invalid operation. Method is not supported yet.
// @param method method that wants to be executed.
error MethodNotSupported(string method);

// File: @openzeppelin/contracts-upgradeable/access/IAccessControlUpgradeable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

pragma solidity ^0.8.0;

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControlUpgradeable {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}

// File: @openzeppelin/contracts/utils/Base64.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (utils/Base64.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides a set of functions to operate with Base64 strings.
 *
 * _Available since v4.5._
 */
library Base64 {
    /**
     * @dev Base64 Encoding/Decoding Table
     */
    string internal constant _TABLE = "ABCDEFGHIJKLMNOPQRSTUVWXYZabcdefghijklmnopqrstuvwxyz0123456789+/";

    /**
     * @dev Converts a `bytes` to its Bytes64 `string` representation.
     */
    function encode(bytes memory data) internal pure returns (string memory) {
        /**
         * Inspired by Brecht Devos (Brechtpd) implementation - MIT licence
         * https://github.com/Brechtpd/base64/blob/e78d9fd951e7b0977ddca77d92dc85183770daf4/base64.sol
         */
        if (data.length == 0) return "";

        // Loads the table into memory
        string memory table = _TABLE;

        // Encoding takes 3 bytes chunks of binary data from `bytes` data parameter
        // and split into 4 numbers of 6 bits.
        // The final Base64 length should be `bytes` data length multiplied by 4/3 rounded up
        // - `data.length + 2`  -> Round up
        // - `/ 3`              -> Number of 3-bytes chunks
        // - `4 *`              -> 4 characters for each chunk
        string memory result = new string(4 * ((data.length + 2) / 3));

        assembly {
            // Prepare the lookup table (skip the first "length" byte)
            let tablePtr := add(table, 1)

            // Prepare result pointer, jump over length
            let resultPtr := add(result, 32)

            // Run over the input, 3 bytes at a time
            for {
                let dataPtr := data
                let endPtr := add(data, mload(data))
            } lt(dataPtr, endPtr) {

            } {
                // Advance 3 bytes
                dataPtr := add(dataPtr, 3)
                let input := mload(dataPtr)

                // To write each character, shift the 3 bytes (18 bits) chunk
                // 4 times in blocks of 6 bits for each character (18, 12, 6, 0)
                // and apply logical AND with 0x3F which is the number of
                // the previous character in the ASCII table prior to the Base64 Table
                // The result is then added to the table to get the character to write,
                // and finally write it in the result pointer but with a left shift
                // of 256 (1 byte) - 8 (1 ASCII char) = 248 bits

                mstore8(resultPtr, mload(add(tablePtr, and(shr(18, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(12, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(shr(6, input), 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance

                mstore8(resultPtr, mload(add(tablePtr, and(input, 0x3F))))
                resultPtr := add(resultPtr, 1) // Advance
            }

            // When data `bytes` is not exactly 3 bytes long
            // it is padded with `=` characters at the end
            switch mod(mload(data), 3)
            case 1 {
                mstore8(sub(resultPtr, 1), 0x3d)
                mstore8(sub(resultPtr, 2), 0x3d)
            }
            case 2 {
                mstore8(sub(resultPtr, 1), 0x3d)
            }
        }

        return result;
    }
}

// File: contracts/structs/LilVillainsStructs.sol



// SPDX-License-Identifier: CC-BY-NC-ND-4.0
pragma solidity ^0.8.9;

struct NFTBaseAttributes {
  uint256 id;
  //INFO: Each position in this arrays represent an attribute
  string[] values;
}

struct NFTBaseAttributesRequest {
  NFTBaseAttributes[] nFTsBaseAttributes;
}

struct Stage {
  string name;
  uint256 price;
  uint32 maxAmount;
  bytes32 root;
  mapping(address => uint32) minters;
  function(uint32, uint32, bytes32[] calldata, bytes calldata) internal beforeMint;
}